
Si bien existen numerosasalternativasparala medición detemperaturaambiente, la medición de lahumedad
relativa ambienteresultaser algo complicada.En amboscasos,lograr una determinadaprecisión implica
disponer desensorescarosy mucho cuidado en lasecciónanalógicay la placa de circuito impreso; hecho que
seve potenciadosi el sensordebeestar aunaciertadistanciadel circuito de procesamiento,y/o en ambientes
hostiles.Los sensorescombinadosde humedady temperaturaambiente SHT-71, desarrollados por lafirma
Sensirion,constan deun par de sensoresy conversoresA/D, circuiteríade calibración,y compensación. La
información espresentadapor unainterfazserie, demodoque el sistema de medición se desentiende del tema
ruidos, linealización analógica,calibración, eimpedancias.Para más detalles,y código ejemplo en C, le
recomendamosreferirse a la CAN-028.En estaoportunidad,desarrollamoscódigoparapoder leerlos sensores
con PIC, bajando notablemente el costo para aplicaciones que requieran gran cantidad de sensores remotos.

Hardware

La conexión es sumamente simple, requieredospines de I/O, uno con capacidadtristateu opencollectorpara
los datos; el otro entrega el clock que controla el timing de esos datos, para ambos sentidos de la comunicación

Software

El fabricante nosproveecódigo en C como paraser compiladoen un procesadorcompatibleMCS51, sin
demasiadoesfuerzo,es posible portarlo a assemblerpara cualquier otraarquitectura,tarea la cual hemos
desarrollado para PIC, en la forma de una pequeña biblioteca de funciones.

El softwarea continuaciónseencargade la lectura de la información provista por el sensor.No realizamos
ningúntipo de procesamiento de lamisma,tareala cualdejamosa criterio del lector.En nuestrocaso,y según
veremosen la CAN-048, reportaremosestevalor a un sistemacentral basadoen Rabbit, el cual hará la
linealización,correccióny presentación delos datos.Paratareaslocalesque norequierangran precisión, el
usuario puede procesar la información que colecta esta nota como un número entero de 16-bits, a su gusto.

Comenzamos por las funciones de lectura y escritura de un byte en el sensor.

; Envía un byte
; Dato en W
; Devuelve resultado en el flag de carry (C)
; C -> Error
s_write_byte:

movwf dat ; guarda el dato
movlw 8 ; 8 bits
movwf bitcntr

swbl1 rlf dat,f ; C = MSB (MSB sale primero)
bcf SCK ; CLK = 0

nop
bcf DAT ; DATA = 0 (open-collector)
bsf STATUS,RP0 ; Bank 1
bsf DATT ; pull-up
btfss STATUS,C ; if C, DATA = pull-up
bcf DATT ; sino DATA = 0
bcf STATUS,RP0 ; Bank 0
bsf SCK ; CLK = 1
decfsz bitcntr,f
goto swbl1
bcf SCK ; CLK = 0
bsf STATUS,RP0 ; Bank 1
bsf DATT ; pull-up
bcf STATUS,RP0 ; Bank 0
bsf SCK ; CLK = 1 (9º clk)
rrf DATPORT,W ; check ack (el SHT pone DATA en 0)
bcf SCK ; CLK = 0
return

; Recibe un byte
; Devuelve el dato en W
s_read_byte:

clrf dat ; DATA = 0
movlw 8 ; 8 bits
movwf bitcntr

srbl1 bsf SCK ; CLK = 1
nop
rrf DATPORT,W ; lee bit en C
rlf dat,f ; y lo lleva al registro (MSB primero)
bcf SCK ; CLK = 0
decfsz bitcntr,f
goto srbl1
bcf DAT ; DATA = 0 (open-collector)
bsf STATUS,RP0 ; Bank 1
btfsc ACK ; si debe hacer ACK
bcf DATT ; DATA=0 (ACK)
bcf STATUS,RP0 ; Bank 0
bsf SCK ; CLK = 1 (9º clk)
nop
bcf SCK ; CLK = 0
bsf STATUS,RP0 ; Bank 1
bsf DATT ; pull-up
bcf STATUS,RP0 ; Bank 0
movf dat,W
return

Como habrándeducido,la selecciónde los pines a emplear serealizamediantemacros, las cuales pueden
redefinirse dentro del programa:

; DAT debe ser el LSB, de otro modo hay que modificar el programa
#define DAT PORTC,0
#define DATPORT PORTC
#define DATT TRISC,0
#define SCK PORTC,1

;                            addr  command  r/w
#define STATUS_REG_W 0x06   ; 000   0011    0
#define STATUS_REG_R 0x07   ; 000   0011    1
#define MEASURE_TEMP 0x03   ; 000   0001    1
#define MEASURE_HUMI 0x05   ; 000   0010    1
#define SHT_RESET    0x1e   ; 000   1111    0

shtvars UDATA_SHR
dat RES 1
bitcntr RES 1
value RES 2
chksum RES 1
flags RES 1

#define ACK flags,0

Luego seguimos por el proceso de inicialización.

; reset: DATA=1 y 9 SCK, seguidos de transstart
;       _____________________________________________________         ________
; DATA:                                                      |_______|
;          _    _    _    _    _    _    _    _    _        ___     ___
; SCK : __| |__| |__| |__| |__| |__| |__| |__| |__| |______|   |___|   |______
; Se debe llamar al inicio o si se observan errores
s_connectionreset:

bcf SCK ; CLK = 0
movlw 9 ; 9 SCK
movwf bitcntr

scrl1 bsf SCK ; CLK = 1 
nop
bcf SCK ; CLK = 0
decfsz bitcntr,f
goto scrl1

                ; sigue transmission start
;genera un transmission start
;       _____         ________
; DATA:      |_______|
;           ___     ___
; SCK : ___|   |___|   |______
s_transstart:

bcf SCK ; CLK = 0
nop
bsf SCK ; CLK = 1
nop
bcf DAT ; DATA = 0 (open-collector)
bsf STATUS,RP0 ; Bank 1
bcf DATT ; DATA=0
bcf STATUS,RP0 ; Bank 0
bcf SCK ; CLK = 0
nop
bsf SCK ; CLK = 1
bsf STATUS,RP0 ; Bank 1
bsf DATT ; pull-up
bcf STATUS,RP0 ; Bank 0
bcf SCK ; CLK = 0
return

A continuación,un juegoderutinasque nos permiteiniciar unamedición,interrogaral sensor elestadode la
misma, y obtener el resultado.

; inicia medición de tmperatura
s_measure_t:

call s_transstart ; transmission start
movlw MEASURE_TEMP
goto s_write_byte ; manda comando al sensor y retorna

; inicia medición de humedad
s_measure_rh:

call s_transstart ; transmission start
movlw MEASURE_HUMI
goto s_write_byte ; manda comando al sensor y retorna

; Llamar hasta que regresa sin carry (C=0), o transcurre demasiado tiempo
s_measure_wait:

rrf DATPORT,W ; lee dato en C
return

; Obtiene resultados
s_measure_get:

bsf ACK ; ACK
call s_read_byte ; lee primer byte (MSB)
movwf value+1
call s_read_byte ; lee segundo byte (LSB)
movwf value
bcf ACK ; no ACK
call s_read_byte ; lee checksum (8-bit)
movwf chksum
return

Dado que el sensor, según la medición de que setrate,puededemorarmás de 200ms enrealizarla,y sabiendo
que nos loindicaráponiendo elpin DATA en estadobajo,lo que haremos esinterrogarloperiódicamente; esta

es larazónpor la cualpartimosla tareaen tressubrutinas,de modoque elprogramaprincipal puedeiniciar
unamedición,realizarsustareasnormalese ir interrogandoperiódicamente al sensor,parafinalmenteobtener
el resultado cuando éste le indica que está disponible:

initsn call s_connectionreset ; inicializa al sensor
; ...
call s_measure_t ; inicia medición de temperatura
btfsc STATUS,C ; verifica si hubo error (reset) o espera
goto hubounerror
;... ; OK, empezó la medición

call s_measure_wait ; pregunta si ya terminó
btfsc STATUS,C ; verifica respuesta
goto hayqueesperar ; espera, verifica si ya esperó demasiado
call s_measure_get ; obtiene resultado
movf value,W ; y lo guarda en su destino
movwf temp
movf value+1,W
movwf temp+1

Lasrutinasdesarrolladasfueron escritas teniendo en cuenta lasespecificacionesdel fabricante,para4MHz de
clock. La cantidadde callsanidadassecomprobóparaun PIC16F630.Paraotras aplicaciones, se sugiere al
interesado revisar que secumplanlos tiemposdetallados en la hoja de datos delSHT-71y la cantidadde calls
permitidos por el hardware stack del PIC utilizado.

El archivo adjunto contiene la totalidad de los listados.

